

SOFTWARE FOR THE ALGORITHMIC WORK WITH ORTHOGONAL POLYNOMIALS AND SPECIAL FUNCTIONS*

WOLFRAM KOEPF†

Abstract. This article was presented as a Plenary Talk at the IWOP 98, Madrid. An introduction to general purpose computer algebra systems is given, and the possibilities to use computer algebra in the work with orthogonal polynomials and special functions are discussed in detail.

Key words. orthogonal polynomials, special functions, computer algebra, Maple V, holonomic functions, differential equations, recurrence equations, hypergeometric functions, Zeilberger's algorithm, Gosper's algorithm, Petkovšek's algorithm.

AMS subject classifications. 33-02, 33C45, 11B37, 34A30, 33C20, 68Q40.

1. Basics. Modern computer algebra systems use heuristics and algorithms for the fast computation with mathematical formulas. For instance, general purpose computer algebra systems like *Axiom* [14], *Derive* [22], *Macsyma* [19], *Maple* [6], *Mathematica* [25] or *Reduce* [13] handle integrals surprisingly well. Even a small system like *Derive* computes *all* explicitly given integrals of Bronshtein and Semedyayev's integral table [4]. The purpose of this paper is to familiarize the reader with some of the underlying techniques which permit computer algebra systems to efficiently solve a large variety of mathematical problems.

We begin by giving examples of important mathematical concepts and methods that are available in general purpose computer algebra systems. Our examples are written in Maple V.5 syntax.

1.1. Linear Algebra. One of the main topics of any computer algebra system is linear algebra. Linear algebra algorithms are used throughout Mathematics; we will see examples in connection with orthogonal polynomials and special functions later.

With Maple, we can compute the solution of a linear system of equations:

```
> solve({x+2*a*y+3*z=4, 5*x+6*y+7*z=8, 9*x+10*y+11*z=12}, {x, y, z});
```

$$\left\{z = \frac{3}{2}, x = \frac{-1}{2}, y = 0\right\}$$

even if parameters are involved. For this purpose, Maple uses a Gauss type algorithm.

Note, that the above system is linear only if a is considered constant. If we consider a as a variable, then a nonlinear system has to be solved:

```
> solve({x+2*a*y+3*z=4, 5*x+6*y+7*z=8, 9*x+10*y+11*z=12}, {a, x, y, z});
```

$$\left\{z = \frac{3}{2}, x = \frac{-1}{2}, y = 0, a = a\right\},$$
$$\left\{z = -\frac{1}{2}y + \frac{3}{2}, a = 1, x = -\frac{1}{2}y - \frac{1}{2}, y = y\right\}.$$

In a forthcoming section, we give more details on nonlinear systems of equations.

Maple has a large linear algebra library:

```
> with(linalg);
```

```
Warning, new definition for norm  
Warning, new definition for trace
```

*Received November 1, 1998. Accepted for publication December 1, 1999. Recommended by F. Marcellán.

†HTWK Leipzig, Department IMN, Gustav-Freytag-Str. 42 A, D-04277 Leipzig, Germany (koepf@imn.htwk-leipzig.de).

[*BlockDiagonal*, *GramSchmidt*, *JordanBlock*, *LUdecomp*, *QRdecomp*, *Wronskian*, *addcol*, *addrow*, *adj*, *adjoint*, *angle*, *augment*, *backsub*, *band*, *basis*, *bezout*, *blockmatrix*, *charmat*, *charpoly*, *cholesky*, *col*, *coldim*, *colspace*, *colspan*, *companion*, *concat*, *cond*, *copyinto*, *crossprod*, *curl*, *definite*, *delcols*, *delrows*, *det*, *diag*, *diverge*, *dotprod*, *eigenvals*, *eigenvalues*, *eigenvectors*, *eigenvects*, *entermatrix*, *equal*, *exponential*, *extend*, *ffgausselim*, *fibonacci*, *forwardsub*, *frobenius*, *gausselim*, *gaussjord*, *geneqns*, *genmatrix*, *grad*, *hadamard*, *hermite*, *hessian*, *hilbert*, *htranspose*, *ihermite*, *indexfunc*, *innerprod*, *intbasis*, *inverse*, *ismith*, *issimilar*, *iszero*, *jacobian*, *jordan*, *kernel*, *laplacian*, *leastsqrs*, *linsolve*, *matadd*, *matrix*, *minor*, *minpoly*, *mulcol*, *mulrow*, *multiply*, *norm*, *normalize*, *nullspace*, *orthog*, *permanent*, *pivot*, *potential*, *randmatrix*, *randvector*, *rank*, *ratform*, *row*, *rowdim*, *rowspan*, *rref*, *scalarmul*, *singularvals*, *smith*, *stackmatrix*, *submatrix*, *subvector*, *sumbasis*, *swapcol*, *swaprow*, *sylvester*, *toeplitz*, *trace*, *transpose*, *vandermonde*, *vecpotent*, *vectdim*, *vector*, *wronskian*]

You can see which procedures are available now. As an example, we compute the determinant of the matrix

$$\begin{pmatrix} 1 & 2a & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$$

by

```
> det([[1, 2*a, 3], [5, 6, 7], [9, 10, 11]]);
      -16 + 16 a
```

and the eigenvalues and eigenvectors for $a = 2$:

```
> eigenvalues([[1, 2, 3], [5, 6, 7], [9, 10, 11]]);
      0, 9 + sqrt(105), 9 - sqrt(105)
> eigenvectors([[1, 2, 3], [5, 6, 7], [9, 10, 11]]);
```

$$\begin{aligned} & [9 + \sqrt{105}, 1, \left\{ \left[\frac{11}{2} - \frac{1}{2} \sqrt{105}, 1, -\frac{7}{2} + \frac{1}{2} \sqrt{105} \right] \right\}], \\ & [9 - \sqrt{105}, 1, \left\{ \left[\frac{11}{2} + \frac{1}{2} \sqrt{105}, 1, -\frac{7}{2} - \frac{1}{2} \sqrt{105} \right] \right\}], [0, 1, \{[1, -2, 1]\}]. \end{aligned}$$

Later we will show how important an efficient implementation of linear algebra can be.

1.2. Polynomial Arithmetic. A second major topic of computer algebra is polynomial arithmetic.

P defines a polynomial

```
> P := (1-x) * sum(x^k, k=0..9);
      P := (1 - x) (1 + x + x^2 + x^3 + x^4 + x^5 + x^6 + x^7 + x^8 + x^9)
```

which is expanded by

```
> expand(P);
      1 - x^10 .
```

Q is a second polynomial

```
> Q := 1 - x^5;
      Q := 1 - x^5
```

and normal cancels common factors of the ratio P/Q :

```
> normal(P/Q);
      x^5 + 1
```

by an application of the Euclidean algorithm.

A highlight of computer algebra is *rational factorization* since the underlying algorithms (factorization modulo a prime and Hensel lifting, or the triple L algorithm, see e.g. [10]) are not suitable for hand computations.

For example, the polynomial P can be factored over \mathbb{Q} by the command

```
> factor(P);
      -(-1 + x)(x + 1)(x^4 + x^3 + x^2 + x + 1)(x^4 - x^3 + x^2 - x + 1)
```

and the following is a rational factorization of $1 - x^{105}$:

```
> factor(1-x^105);
      -(-1 + x)(x^6 + x^5 + x^4 + x^3 + x^2 + x + 1)(x^4 + x^3 + x^2 + x + 1)(1 - x + x^5
      - x^6 + x^7 - x^8 + x^10 - x^11 + x^12 - x^13 + x^14 - x^16 + x^17 - x^18 + x^19
      - x^23 + x^24)(x^2 + x + 1)(1 - x + x^3 - x^4 + x^6 - x^8 + x^9 - x^11 + x^12)
      (1 - x + x^3 - x^4 + x^5 - x^7 + x^8)(1 + x + x^2 - x^5 - x^6 - 2x^7 - x^8 - x^9
      - x^24 + x^12 + x^13 + x^14 + x^16 + x^17 + x^15 + x^48 - x^20 - x^22 - x^26 - x^28
      + x^31 + x^32 + x^33 + x^34 + x^35 + x^36 - x^39 - x^40 - 2x^41 - x^42 - x^43 + x^46
      + x^47)
```

Note that 105 is the smallest exponent such that the rational factorization of $1 - x^n$ contains coefficients different from 0 or ± 1 .

Next, we define a multivariate polynomial

```
> Product(x^(2*k-1)-y^k/k^2, k=1..7);
```

$$\prod_{k=1}^7 \left(x^{2k-1} - \frac{y^k}{k^2} \right)$$

whose expanded form is a huge expression

```
> term:=expand(product(x^(2*k-1)-y^k/k^2, k=1..7));
```

$$\begin{aligned} \text{term} := & -\frac{1}{25}x^{40}y^5 + \frac{1}{1225}x^{27}y^{12} + \frac{421}{176400}x^{29}y^{11} - \frac{1}{44100}x^{16}y^{18} - \frac{1}{16}x^{42}y^4 \\ & + \frac{113}{28224}x^{31}y^{10} - \frac{1}{28224}x^{18}y^{17} + \frac{16969}{1587600}x^{33}y^9 - \frac{181}{1587600}x^{20}y^{16} \\ & - \frac{71}{235200}x^{22}y^{15} + \frac{1}{705600}x^9y^{22} + \frac{5609}{176400}x^{35}y^8 - \frac{1}{9}x^{44}y^3 - \frac{47}{45360}x^{24}y^{14} \\ & + \frac{1}{396900}x^{11}y^{21} + \frac{161}{3600}x^{37}y^7 - \frac{5099}{3175200}x^{26}y^{13} + \frac{61}{6350400}x^{13}y^{20} \\ & - \frac{2069}{564480}x^{28}y^{12} + \frac{1}{26880}x^{15}y^{19} + \frac{13}{181440}x^{17}y^{18} - \frac{1}{6350400}x^4y^{25} \\ & - \frac{1}{4}x^{46}y^2 + \frac{89}{1600}x^{39}y^6 - \frac{17147}{3175200}x^{30}y^{11} + \frac{559}{3628800}x^{19}y^{17} \\ & - \frac{1}{2822400}x^6y^{24} + \frac{13}{144}x^{41}y^5 - \frac{18733}{1587600}x^{32}y^{10} + \frac{167}{453600}x^{21}y^{16} \\ & - \frac{13}{6350400}x^8y^{23} - \frac{21}{1600}x^{34}y^9 + \frac{2629}{5080320}x^{23}y^{15} - \frac{89}{25401600}x^{10}y^{22} \\ & + \frac{1091}{1270080}x^{25}y^{14} - \frac{1}{90720}x^{12}y^{21} - \frac{209}{12700800}x^{14}y^{20} + \frac{1}{25401600}xy^{27} \\ & - yx^{48} + \frac{1}{9}y^4x^{43} - \frac{61}{3600}y^8x^{36} + \frac{401}{313600}y^{13}x^{27} - \frac{23}{635040}y^{19}x^{16} \\ & + \frac{1}{6350400}y^{26}x^3 + \frac{1}{4}y^3x^{45} - \frac{1}{64}y^7x^{38} + \frac{181}{129600}y^{12}x^{29} - \frac{113}{2822400}y^{18}x^{18} \end{aligned}$$

$$\begin{aligned}
 & + \frac{1}{2822400} y^{25} x^5 - \frac{1}{36} y^6 x^{40} + \frac{1}{900} y^{11} x^{31} - \frac{421}{6350400} y^{17} x^{20} \\
 & + \frac{1}{1587600} y^{24} x^7 + \frac{1}{576} y^{10} x^{33} - \frac{1}{20736} y^{16} x^{22} + \frac{1}{1016064} y^{23} x^9 - \frac{1}{36} x^{38} y^6 \\
 & + \frac{1}{1764} x^{25} y^{13} - \frac{1}{14400} y^{15} x^{24} + \frac{1}{705600} y^{22} x^{11} - \frac{1}{49} x^{36} y^7 \\
 & + \frac{1}{518400} y^{21} x^{13} + x^{49} - \frac{1}{25401600} y^{28}
 \end{aligned}$$

which is a polynomial of degree 49 in x and of degree 28 in y .

It is beautiful (and will turn out to be essential in the sequel) that computer algebra systems have no problems factoring such expressions over the rationals in reasonable time:

> factor(term);

$$\frac{1}{25401600} (x-y)(-y^2+4x^3)(-y^3+9x^5)(-y^5+25x^9)(-y^6+36x^{11}) \\
 (49x^{13}-y^7)(-y^4+16x^7)$$

1.3. Polynomial Systems. We return to the problem of nonlinear systems of equations. Buchberger's algorithm is an extension of Gaussian elimination to the multivariate case. It constitutes—given a certain term order—an elimination scheme to find a normal form for a given polynomial system, which can be used to find the general solution of a nonlinear system.

We consider the following system of equations:

```

> LIST := {9*B*A+4*d-6*c*d=0,
>         -9*a*b+9*B*A=0,
>         -18*B*A+12*d-12*c*d+4*d^2=0,
>         6*b*d-36*a*b+2*d+6*a*d=0,
>         -4*d^2+12*b*d-36*a*b+12*a*d=0,
>         -8*C-9+9*B+9*A-4*d+12*c=0,
>         -8*d-7+12*c+27*A+27*B-24*C=0,
>         8-3*a-3*b-32*C+27*A+27*B=0,
>         6-16*C+18*A+18*B-12*a+4*d-12*b=0,
>         4-12*a-12*b+8*d=0,
>         -C-2+3*c=0};

```

$$\begin{aligned}
 LIST := \{ & 9BA + 4d - 6cd = 0, -9ab + 9BA = 0, \\
 & -18BA + 12d - 12cd + 4d^2 = 0, 6bd - 36ab + 2d + 6ad = 0, \\
 & -4d^2 + 12bd - 36ab + 12ad = 0, 4 - 12a - 12b + 8d = 0, \\
 & -C - 2 + 3c = 0, -8C - 9 + 9B + 9A - 4d + 12c = 0, \\
 & -8d - 7 + 12c + 27A + 27B - 24C = 0, \\
 & 8 - 3a - 3b - 32C + 27A + 27B = 0, \\
 & 6 - 16C + 18A + 18B - 12a + 4d - 12b = 0 \}.
 \end{aligned}$$

The solve command gives the general solution:

```
> solve(LIST, {A, B, C, a, b, c, d});
```

$$\begin{aligned}
 & \{c = \frac{5}{6} + \frac{1}{6}d, A = \frac{1}{3}d + \frac{1}{3}, B = \frac{1}{3}d, b = \frac{1}{3}d, a = \frac{1}{3}d + \frac{1}{3}, C = \frac{1}{2} + \frac{1}{2}d, d = d\}, \\
 & \{c = \frac{5}{6} + \frac{1}{6}d, A = \frac{1}{3}d + \frac{1}{3}, B = \frac{1}{3}d, b = \frac{1}{3}d + \frac{1}{3}, a = \frac{1}{3}d, C = \frac{1}{2} + \frac{1}{2}d, \\
 & d = d\}, \{c = \frac{5}{6} + \frac{1}{6}d, b = \frac{1}{3}d, B = \frac{1}{3}d + \frac{1}{3}, A = \frac{1}{3}d, a = \frac{1}{3}d + \frac{1}{3}, \\
 & C = \frac{1}{2} + \frac{1}{2}d, d = d\}, \{c = \frac{5}{6} + \frac{1}{6}d, b = \frac{1}{3}d + \frac{1}{3}, B = \frac{1}{3}d + \frac{1}{3}, A = \frac{1}{3}d,
 \end{aligned}$$

$$a = \frac{1}{3}d, C = \frac{1}{2} + \frac{1}{2}d, d = d\} .$$

In an application, we will see this example later.

1.4. Differentiation and Integration. Differentiation is done using the differentiation rules. This is an easy task. For the example function

```
> input:=exp(x-x^2)*sin(x^6-1);
      input := e(x-x2) sin(x6 - 1)
```

obviously the product rule is used:

```
> derivative:=diff(input,x);
      derivative := (1 - 2x) e(x-x2) sin(x6 - 1) + 6 e(x-x2) cos(x6 - 1) x5 .
```

Integration is *much* more difficult, and the different systems have different approaches. Whereas Derive uses a good collection of heuristics which enable the system to compute all explicitly given integrals of Bronshtein and Semydayev's integral table [4], as already mentioned, Maple uses an algorithmic approach.

In the sixties Risch developed an algorithm to compute an *elementary* antiderivative whenever one exists. If no such antiderivative exists, his algorithm returns this information. Here elementary means that both integrand and antiderivative are rationally composed of exponentials and logarithms (see e.g. [5]). Adjoining the complex unit i (denoted in Maple by I), trigonometric functions can be treated as well.

We integrate the derivative above. This takes a little longer:

```
> integral:=int(derivative,x);
      integral := -\frac{1}{2} I e^{((x-1)(Ix^5+Ix^4+Ix^3+Ix^2-x+Ix+I))}
      + \frac{1}{2} I e^{-(x-1)(Ix^5+Ix^4+Ix^3+Ix^2+x+Ix+I)} .
```

Since i is adjoined, the resulting function looks not very familiar although it is algebraically equivalent to our input function. In this particular case, we can convert both functions to the same *normal form* by first converting exponentials to trigonometrics and applying then rational factorization:

```
> factor(convert(integral,trig));
      -sin((x-1)(x+1)(x^2+x+1)(x^2-x+1))(-cosh(x(x-1))+sinh(x(x-1)))
> factor(convert(input,trig));
      -sin((x-1)(x+1)(x^2+x+1)(x^2-x+1))(-cosh(x(x-1))+sinh(x(x-1)))
```


Note, however, that one can prove that for general transcendental expressions a normal form does not exist.

1.5. Differential Equations. In engineering and in natural sciences the symbolic and numeric solution of differential equations is rather important. We enter an ordinary differential equation:

```
> DE:=diff(y(x),x)=1+y(x)^2;
      DE := \frac{\partial}{\partial x} y(x) = 1 + y(x)^2 .
```


After loading the `DEtools` package, we can use the procedure `dfieldplot` to plot a direction field of the differential equation:

```
> with(DEtools):
> dfieldplot(DE, y(x), x=-5..5, y=-5..5);
```



Given an initial value, the command `DEplot` plots a numeric solution by a Runge-Kutta type approach:

```
> DEplot({DE}, {y(x)}, x=-1..1, [[y(0)=0]]);
```



Using a combination of heuristic and algorithmic techniques, Maple can solve many ordinary differential equations explicitly. Our initial value problem has the solution:

```
> dsolve({DE, y(0)=0}, y(x));
```

$$y(x) = \tan(x) .$$

As another example, we consider a linear differential equation of second order.

> DE:=diff(y(x),x\$2)-y(x)=sin(x)*x;

$$DE := \left(\frac{\partial^2}{\partial x^2} y(x)\right) - y(x) = \sin(x) x$$


with explicit solution

> dsolve(DE,y(x));

$$\begin{aligned} y(x) = & \left(\frac{1}{2}\left(-\frac{1}{2}x + \frac{1}{2}\right)e^x \cos(x) + \frac{1}{4}\sin(x)xe^x + \frac{1}{2}\left(-\frac{1}{2}x - \frac{1}{2}\right)e^{(-x)}\cos(x)\right. \\ & - \frac{1}{4}xe^{(-x)}\sin(x)\sinh(x) + \left(-\frac{1}{2}\left(-\frac{1}{2}x + \frac{1}{2}\right)e^x \cos(x) - \frac{1}{4}\sin(x)xe^x\right. \\ & + \frac{1}{2}\left(-\frac{1}{2}x - \frac{1}{2}\right)e^{(-x)}\cos(x) - \frac{1}{4}xe^{(-x)}\sin(x)\cosh(x) + _C1\sinh(x) \\ & \left. + _C2\cosh(x)\right) . \end{aligned}$$

A plot based on a numerical computation is given by

> DEplot({DE},{y(x)},x=-5..5,[y(0)=0,D(y)(0)=1]);



The corresponding initial value problem has the explicit solution

> solution:=dsolve({DE,y(0)=0,D(y)(0)=1},y(x));

$$\begin{aligned} solution := y(x) = & \left(\frac{1}{2}\left(-\frac{1}{2}x + \frac{1}{2}\right)e^x \cos(x) + \frac{1}{4}\sin(x)xe^x\right. \\ & + \frac{1}{2}\left(-\frac{1}{2}x - \frac{1}{2}\right)e^{(-x)}\cos(x) - \frac{1}{4}xe^{(-x)}\sin(x)\sinh(x) + \left. \left(-\frac{1}{2}\left(-\frac{1}{2}x + \frac{1}{2}\right)e^x \cos(x) - \frac{1}{4}\sin(x)xe^x + \frac{1}{2}\left(-\frac{1}{2}x - \frac{1}{2}\right)e^{(-x)}\cos(x)\right.\right. \\ & \left. - \frac{1}{4}xe^{(-x)}\sin(x)\cosh(x) + \sinh(x) + \frac{1}{2}\cosh(x)\right) \end{aligned}$$

which can be simplified to

```
> simplify(convert(rhs(solution), trig));
```

$$\sinh(x) + \frac{1}{2} \cosh(x) - \frac{1}{2} \cos(x) - \frac{1}{2} \sin(x) x .$$

1.6. Formal Power Series and Differential Equations. Next, we consider the opposite problem to generate differential equations from expressions. This will lead us also to the generation of power series of hypergeometric type.

After loading the FPS package [12]

```
> with(share): with(FPS):
```

See ?share and ?share,contents for information about the share library
Share Library: FPS

Author: Gruntz, Dominik.

Description: FPS function attempts to find a formal power series expansion for a function in terms of a formula for the coefficients

we can, for example, compute the formal power series of the square of the inverse tangent function:

```
> FPS(arcsin(x)^2, x);
```

$$(1.1) \quad \sum_{k=0}^{\infty} \frac{(k!)^2 4^k x^{(2k+2)}}{(k+1)(1+2k)!} .$$

The algorithm behind this procedure is outlined below (see [16], [12]):

In the first step, by linear algebra techniques, a homogeneous linear differential equation with polynomial coefficients is sought for the given expression

```
> DE:=SimpleDE(arcsin(x)^2, x, F);
```

$$DE := (x-1)(x+1) \left(\frac{\partial^3}{\partial x^3} F(x) \right) + \left(\frac{\partial}{\partial x} F(x) \right) + 3x \left(\frac{\partial^2}{\partial x^2} F(x) \right) = 0 .$$

We call such a differential equation as well as the corresponding function *holonomic*. Next, substituting the series

$$F(x) = \sum_{k=0}^{\infty} a_k x^k$$

in this differential equation and equating coefficients yields the holonomic recurrence equation for a_k :

```
> RE:=SimpleRE(arcsin(x)^2, x, a);
```

$$RE := -(k+1)(a(k+3)k^2 - k^2 a(k+1) + 5a(k+3)k - 2ka(k+1) - a(k+1) + 6a(k+3)) = 0$$

which can be put in factored form

```
> map(factor, collect(lhs(RE), a))=0;
```

$$(1.2) \quad -(k+1)(k+2)(k+3)a(k+3) + (k+1)^3 a(k+1) = 0$$

Notice that the resulting recurrence equation gives a_{k+2} as a rational multiple of a_k . If A_{k+1} is a rational multiple of A_k then it is called a *hypergeometric term*. From (1.2), a_k can be easily computed using two initial values. This finally generates the explicit series representation (1.1). Note, however, that for an explicit representation the above factorization is necessary; see (2.1).

By solving the differential equation for $F(x) = \arcsin^2(x)$ with two initial values, we would like to reconstruct the input:


```

> solution:=dsolve({DE,F(0)=0,D(F)(0)=0,(D@@2)(F)(0)=2},F(x));

solution :=
F(x) = 1/4 pi^2 + I pi ln(x + sqrt((x-1)(x+1))) - ln(x + sqrt((x-1)(x+1)))^2

> convert(arcsin(x)^2,ln);
      -ln(sqrt(1-x^2)+Ix)^2

```

As before, we see that transcendental functions come in quite different disguises.

It turns out that sum and product of two holonomic functions are again holonomic, and the corresponding holonomic (differential or recurrence) equations can be constructed from the given holonomic equations by linear algebra (see [1], [2], [24], [23]).

As an example, we consider both the sum and the product of the functions $f(x) = \arcsin x$ and $g(x) = e^x$. Here are their holonomic equations:

```

> DE1:=SimpleDE(arcsin(x),x,F);
      DE1 := (x-1)(x+1)(d^2 F(x)/dx^2) + (d F(x)/dx)x = 0

> DE2:=SimpleDE(exp(x),x,F);
      DE2 := (d F(x)/dx) - F(x) = 0.

```

From these, we can compute the holonomic equations that are valid for $f(x) + g(x)$ and $f(x) \cdot g(x)$. For this purpose, we load the `gfun` package [23]:

```

> with(gfun);

[Laplace, algebraicsubs, algeqtodiffeq, algeqtoseries, algfuntoalgeq, borel,
cauchyproduct, diffeq*diffeq, diffeq+diffeq, diffeqtorec, guesseqn,
guessgf, hadamardproduct, holexprtodiffeq, invborel, listtoalgeq,
listtodiffeq, listtohypergeom, listtolist, listtoratpoly, listtorec,
listtoseries, listtoseries/Laplace, listtoseries/egf, listtoseries/lgdegf,
listtoseries/lgdogf, listtoseries/ogf, listtoseries/revegf,
listtoseries/revogf, maxdegcoeff, maxdegeqn, maxordereqn,
mindegcoeff, mindegeqn, minordereqn, optionsgf, poltodiffeq,
poltorec, ratpolytcoeff, rec*rec, rec+rec, rectodiffeq, rectoproc,
seriestoalgeq, seriestodiffeq, seriestohypergeom, seriestolist,
seriestoratpoly, seriestorec, seriestoseries]

```

The procedures `'diffeq+diffeq'` and `'diffeq*diffeq'` compute the differential equations of sum and product, respectively:

```

> 'diffeq+diffeq'(DE1,DE2,F(x));
      {(-x^3-2x^2+x-1)D(F)(x)+(-x^4+4x^2)(D^(2))(F)(x)
      +(1-2x^2+x^4-x+x^3)(D^(3))(F)(x),(D^(2))(F)(0)=-C0}

> 'diffeq*diffeq'(DE1,DE2,F(x));
      (-1+x^2-x)F(x)+(x+2-2x^2)D(F)(x)+(-1+x^2)(D^(2))(F)(x)

```

which we could also have obtained using `SimpleDE` directly:¹

```

> SimpleDE(arcsin(x)+exp(x),x,F);

```

¹Note that `SimpleDE` uses a slightly different approach (also based on linear algebra) that sometimes can find differential equations of lower order than `'diffeq+diffeq'` and `'diffeq*diffeq'`.

$$(x-1)(x+1)(x-1+x^2)\left(\frac{\partial^3}{\partial x^3}F(x)\right) + (x-x^3-2x^2-1)\left(\frac{\partial}{\partial x}F(x)\right) - x^2(x-2)(x+2)\left(\frac{\partial^2}{\partial x^2}F(x)\right) = 0$$

> SimpleDE(arcsin(x)*exp(x), x, F);

$$(x-1)(x+1)\left(\frac{\partial^2}{\partial x^2}F(x)\right) + (x+2-2x^2)\left(\frac{\partial}{\partial x}F(x)\right) + (-1+x^2-x)F(x) = 0.$$

SimpleDE can also generate differential equations for some special functions, e.g., for the Bessel functions $J_n(x)$:

> DE:=SimpleDE(BesselJ(n, x), x, F);

$$DE := \left(\frac{\partial^2}{\partial x^2}F(x)\right)x^2 - (n-x)(n+x)F(x) + \left(\frac{\partial}{\partial x}F(x)\right)x = 0.$$

Maple can solve this differential equation easily:

> dsolve(DE, F(x));

$$F(x) = _C1 \text{BesselY}(n, x) + _C2 \text{BesselJ}(n, x).$$

Even the more complicated differential equation of the product

> DE:=SimpleDE(BesselJ(n, x)*exp(x), x, F);

$$DE := (2x^2 - n^2 - x)F(x) - (-1 + 2x)x\left(\frac{\partial}{\partial x}F(x)\right) + \left(\frac{\partial^2}{\partial x^2}F(x)\right)x^2 = 0$$

can be treated by Maple

> dsolve(DE, F(x));

$$F(x) = _C1 \text{BesselJ}(n, x)e^x + _C2 \text{BesselY}(n, x)e^x,$$

but for the differential equation

> DE:=SimpleDE(BesselJ(n, x)+exp(x), x, F);

$$DE := (2x^4 - 3x^2n^2 + x^3 - 3n^2x - x^2 + n^4 - n^2)F(x) + (-n^4 + 3x^2 + n^2 + x^3 - 2x^4 + 3x^2n^2)\left(\frac{\partial}{\partial x}F(x)\right) - (-2x^3 + n^2x + x^2 - 3n^2 + 2x)x\left(\frac{\partial^2}{\partial x^2}F(x)\right) + x^2(-2x^2 + n^2 - x)\left(\frac{\partial^3}{\partial x^3}F(x)\right) = 0,$$

Maple fails

> dsolve(DE, F(x));

$$F(x) = _C1 e^x + e^x \text{DESol} \left(\left\{ -Y(x) - \frac{x(2x+1)\left(\frac{\partial}{\partial x} - Y(x)\right)}{-2x^2 + n^2 - x} - \frac{(-2x^4 + x^2n^2 - x^3)\left(\frac{\partial^2}{\partial x^2} - Y(x)\right)}{(-2x^2 + n^2 - x)^2} \right\}, \{-Y(x)\} \right)$$

although Maple was able to find the exponential summand (and hence reduced the order by one).

Nevertheless, it is not astonishing that Maple cannot find all such solutions since for this type of nonelementary solutions no algorithms exist.

2. Special Functions and Computer Algebra. Power series of hypergeometric type—the Bessel functions are also of this type—are the most important special functions.

The *generalized hypergeometric series* is given by

$$(2.1) \quad {}_pF_q \left(\begin{matrix} a_1 & a_2 & \cdots & a_p \\ b_1 & b_2 & \cdots & b_q \end{matrix} \middle| x \right) := \sum_{k=0}^{\infty} A_k x^k = \sum_{k=0}^{\infty} \frac{(a_1)_k \cdot (a_2)_k \cdots (a_p)_k}{(b_1)_k \cdot (b_2)_k \cdots (b_q)_k k!} x^k$$

where $(a)_k := \prod_{j=1}^k (a+j-1) = \Gamma(a+k)/\Gamma(a)$ denotes the *Pochhammer-Symbol* or *shifted factorial*.

A_k is a hypergeometric term and fulfills the recurrence equation ($k \in \mathbb{N}$)

$$A_{k+1} = \frac{(k+a_1) \cdots (k+a_p)}{(k+b_1) \cdots (k+b_q)(k+1)} \cdot A_k$$

with the initial value

$$A_0 = 1.$$

In Maple the hypergeometric series is given as `hypergeom(plist, qlist, x)`, where

$$\text{plist} = [a_1, a_2, \dots, a_p], \quad \text{and} \quad \text{qlist} = [b_1, b_2, \dots, b_q].$$

Here are some more hypergeometric examples:

> `F:=sqrt(x)*arcsin(sqrt(x))+sqrt(1-x);`
 $F := \sqrt{x} \arcsin(\sqrt{x}) + \sqrt{1-x}$

> `SUM:=FPS(F,x);`

$$SUM := \sum_{k=0}^{\infty} \frac{4^{(-k)} (2k)! x^k}{(k!)^2 (2k-1)^2}$$

> `convert(SUM,hypergeom);`

$$\text{hypergeom}\left(\left[\frac{-1}{2}, \frac{-1}{2}\right], \left[\frac{1}{2}\right], x\right)$$

> `F:=-sqrt(Pi)/2*sqrt(x)*erf(sqrt(x))*(1+1/2/x)+exp(-x)/2;`

$$F := -\frac{1}{2} \sqrt{\pi} \sqrt{x} \operatorname{erf}(\sqrt{x}) \left(1 + \frac{1}{2} \frac{1}{x}\right) - \frac{1}{2} e^{(-x)}$$

> `SUM:=FPS(F,x);`

$$SUM := \sum_{k=0}^{\infty} \frac{(-1)^k x^k}{k! (2k+1) (-1+2k)}$$

> `convert(SUM,hypergeom);`

$$-\text{KummerM}\left(\frac{-1}{2}, \frac{3}{2}, -x\right)$$

With `convert`, one can convert series into hypergeometric notation; `KummerM` is another name for the confluent hypergeometric function ${}_1F_1$.

2.1. Summation. Whereas the `FPS` command converts expressions into series representations, the opposite question is to find explicit representations for sums. Note that the remaining examples in this paper are taking from [17].

The main interest lies in sums of hypergeometric terms. As an example, we ask: Why does Maple evaluate the sum

> `sum((-1)^k*binomial(n,k),k=a..b);`

$$-\frac{(b+1)(-1)^{(b+1)} \operatorname{binomial}(n, b+1)}{n} + \frac{a(-1)^a \operatorname{binomial}(n, a)}{n}$$

for arbitrary bounds a and b , but fails with

```
> sum(binomial(n,k),k);
      
$$\sum_k \text{binomial}(n, k) ?$$

```

On the other hand, for the special bounds $a = 0$ and $b = n$, Maple is successful, again:

```
> sum(binomial(n,k),k=0..n);
      
$$2^n .$$

```

The reason for this behavior is that the first summand, $(-1)^k \binom{n}{k}$, has a hypergeometric term antidifference (in the variable k), and the second, $\binom{n}{k}$, has not. The last sum is a definite sum with natural bounds, i.e., the sum can be considered as infinite sum ($k = -\infty.. \infty$), and the result, again, is a hypergeometric term (in the variable n). We will see how we can find these types of results algorithmically.

Note that s_k is called an *antidifference* of a_k , if

$$s_{k+1} - s_k = a_k ,$$

If such an antidifference is known, then summation is trivial since by telescoping

$$\sum_{k=a}^b a_k = (s_{b+1} - s_b) + (s_b - s_{b-1}) + \cdots + (s_{a+1} - s_a) = s_{b+1} - s_a .$$

This is very similar to the integration case.

The antidifference of the first summand is given by

```
> sum((-1)^k*binomial(n,k),k);
      
$$-\frac{k(-1)^k \text{binomial}(n, k)}{n}$$

```

We can increase the level of user information by the command

```
> infolevel[sum]:=3:
```

Let's try to prove the statement

$$\sum_{k=1}^{\infty} \frac{(-1)^{k+1} (4k+1) (2k)!}{k! 4^k (2k-1) (k+1)!} = 1$$

that was posed in SIAM Review 36, 1994, Problem 94-2 [21]. We compute an antidifference

```
> summand:=(-1)^(k+1)*(4*k+1)*(2*k)!/(k!*4^k*(2*k-1)*(k+1)!);
> sum(summand,k);
```

```
sum/indefnew:  indefinite summation
sum/extgosp:   applying Gosper algorithm to a( k ):=
(-1)^(k+1)*(4*k+1)*(2*k)!/k!/(4^k)/(2*k-1)/(k+1)!
sum/gosp:     a( k )/a( k -1):=
-1/2*(4*k+1)/(4*k-3)/(k+1)*(2*k-3)
sum/gospnew:  Gosper's algorithm applicable
sum/gospnew:  p:= 4*k+1
sum/gospnew:  q:= -2*k+3
sum/gospnew:  r:= 2*k+2
sum/gospnew:  degreebound:= 0
sum/gospnew:  solving equations to find f
sum/gospnew:  Gosper's algorithm successful
sum/gospnew:  f:= -1
sum/indefnew:  indefinite summation finished
      
$$-2 \frac{(k+1)(-1)^{(k+1)}(2k)!}{k! 4^k (2k-1) (k+1)!} ,$$

```

with success. Taking the limit as $n \rightarrow \infty$, one gets therefore

```
> sum(summand,k=1..infinity);
sum/infinite: infinite summation
1.
```

Moreover, from the user information we see that *Gosper's algorithm* is applied.

If a_k is a hypergeometric term, i.e., if²

$$\frac{a_{k+1}}{a_k} = \frac{b_k}{c_k} \in \mathbb{Q}(k),$$

then Gosper's algorithm *decides* whether or not the antidifference s_k is a hypergeometric term, and computes it in the affirmative case. In detail: Given³

$$\frac{a_{k+1}}{a_k} = \frac{b_k}{c_k}, \quad b_k, c_k \in \mathbb{Q}[k],$$

a representation

$$\frac{b_k}{c_k} = \frac{p_{k+1}}{p_k} \frac{q_{k+1}}{r_{k+1}}, \quad p_k, q_k, r_k \in \mathbb{Q}[k]$$

is computed for which

$$\gcd(q_k, r_{k+j}) = 1 \quad \text{for all } j \in \mathbb{N}_0.$$

This can be done by a resultant computation [11] or by rational factorization (see [17], [18], [20]).

The essential fact is then: f_k , defined by

$$s_k = \frac{r_k}{p_k} f_{k-1} a_k$$

is rational, and the above gcd-condition yields even $f_k \in \mathbb{Q}[k]$, and f_k satisfies the inhomogeneous recurrence equation

$$p_k = q_{k+1} f_k - r_k f_{k-1}.$$

After calculating the degree of f_k , only linear algebra is needed to compute f_k . The output of the procedure is either

$$s_k = \frac{r_k}{p_k} f_{k-1} a_k$$

or the statement “*There is no elementary (= hypergeometric term) antidifference*”.

In [17], many algorithms that are connected with Gosper's, are treated in detail and their Maple implementations are given.

```
After loading 'hsum.mpl',4
> read('hsum.mpl');
```

Copyright 1998 Wolfram Koepf, Konrad – Zuse – Zentrum Berlin

² $\mathbb{Q}(k)$: rational functions over \mathbb{Q} .

³ $\mathbb{Q}[k]$: polynomials over \mathbb{Q} .

⁴The packages 'hsum.mpl' and 'qsum.mpl' can be obtained from the URL www.imn.htwk-leipzig.de/~koepf/research.html.

we can repeat the above calculation with the command

```
> gosper((-1)^(k+1)*(4*k+1)*(2*k)!/(k!*4^k*(2*k-1)*(k+1)!),k);
```

$$-2 \frac{(k+1)(-1)^{(k+1)}(2k)!}{k!4^k(2k-1)(k+1)!}.$$

The computation

```
> gosper(1/k,k);
```

Error, (in gosper) no hypergeometric term antidifference exists
is not worthless since it proves that the harmonic numbers

$$H_n := \sum_{k=1}^n \frac{1}{k}$$

do *not* constitute a hypergeometric term, corresponding to the fact that the logarithmic function

$$\ln x = \int_1^x \frac{1}{t} dt$$

cannot be written in terms of exponentials.

Zeilberger's algorithm is an extension of Gosper's for definite sums. It generates, for example, the right-hand sides of the identities

$$\sum_{k=0}^n \binom{n}{k} = 2^n,$$

$$\sum_{k=0}^n \binom{n}{k}^2 = \frac{(2n)!}{(n!)^2}$$

by the commands

```
> closedform(binomial(n,k),k,n);
```

$$2^n$$

```
> closedform(binomial(n,k)^2,k,n);
```

$$\frac{(2n)!}{(n!)^2}.$$

Here are the details: If $F(n, k)$ is a hypergeometric term in n and k , i.e.

$$\frac{F(n+1, k)}{F(n, k)} \quad \text{and} \quad \frac{F(n, k+1)}{F(n, k)} \in \mathbb{Q}(n, k),$$

then Zeilberger's algorithm generates a *holonomic recurrence equation* for

$$s_n := \sum_{k \in \mathbb{Z}} F(n, k).$$

This is performed by starting with $J = 1$ and iterating if necessary: Set

$$a_k := F(n, k) + \sum_{j=1}^J \sigma_j(n) F(n+j, k)$$

with as yet undetermined variables σ_j . Apply Gosper's algorithm to a_k . In the last step, solve the linear system at the same time for the coefficients of f_k and the variables σ_j ($j = 1, \dots, J$). In the affirmative case, this yields

$$G(n, k+1) - G(n, k) = a_k .$$

Output: By summation one gets

$$s_n + \sum_{j=1}^J \sigma_j(n) s_{n+j} = 0 .$$

If the resulting recurrence equation is of first order, then s_n is a hypergeometric term, given by formula (2.1).

We would like to point out that the most time consuming part of Zeilberger's algorithm is the last step which is to solve a linear system. This linear system, however, often has many variables, and its coefficients are polynomials or rational functions. Here, an efficient implementation of linear algebra is important. Furthermore, the resulting recurrence equation usually needs factored coefficients because otherwise the results look unnecessarily complicated. We will see such a situation soon.

We give some examples: Each of the following series represents the *Legendre polynomials*:

$$\begin{aligned}
 P_n(x) &= \sum_{k=0}^n \binom{n}{k} \binom{-n-1}{k} \left(\frac{1-x}{2}\right)^k \\
 &= {}_2F_1\left(\begin{matrix} -n, n+1 \\ 1 \end{matrix} \middle| \frac{1-x}{2}\right) \\
 &= \frac{1}{2^n} \sum_{k=0}^n \binom{n}{k}^2 (x-1)^{n-k} (x+1)^k \\
 &= \left(\frac{1-x}{2}\right)^n {}_2F_1\left(\begin{matrix} -n, -n \\ 1 \end{matrix} \middle| \frac{1+x}{1-x}\right) \\
 &= \frac{1}{2^n} \sum_{k=0}^{\lfloor n/2 \rfloor} (-1)^k \binom{n}{k} \binom{2n-2k}{n} x^{n-2k} \\
 &= \binom{2n}{n} \left(\frac{x}{2}\right)^n {}_2F_1\left(\begin{matrix} -n/2, -n/2+1/2 \\ -n+1/2 \end{matrix} \middle| \frac{1}{x^2}\right) \\
 &= x^n {}_2F_1\left(\begin{matrix} -n/2, -n/2+1/2 \\ 1 \end{matrix} \middle| 1 - \frac{1}{x^2}\right) .
 \end{aligned}$$

Again, you see, that functions come in quite different disguises. How can we show that these systems define the same family of functions? Zeilberger's paradigm is to show that they satisfy the same recurrence equation, then it is sufficient to check a finite number of initial values.

Here are the recurrence equations for the different sums:

> P := 'P' :

```

> sumrecursion(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k,P(n));
      (n+2)P(n+2) - (2n+3)xP(n+1) + (n+1)P(n) = 0
> sumrecursion(1/2^n*binomial(n,k)^2*(x-1)^(n-k)*
(x+1)^k,k,P(n));
      (n+2)P(n+2) - (2n+3)xP(n+1) + (n+1)P(n) = 0
> sumrecursion(1/2^n*(-1)^k*binomial(n,k)*
binomial(2*n-2*k,n)*x^(n-2*k),k,P(n));
      (n+2)P(n+2) - (2n+3)xP(n+1) + (n+1)P(n) = 0
> sumrecursion(x^n*hyperterm([-n/2,-n/2+1/2],[1],1-1/x^2,k),k,P(n));
      (n+2)P(n+2) - (2n+3)xP(n+1) + (n+1)P(n) = 0

```

We omit the computation of the initial values.

The `Sumtohyper` procedure of the `hsum` package is slightly more efficient than `convert/hypergeom` by converting a series into hypergeometric notation.

```

> Sumtohyper(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k);
      Hypergeom([n+1, -n], [1], 1/2 - 1/2 x)
> Sumtohyper(1/2^n*binomial(n,k)^2*(x-1)^(n-k)*(x+1)^k,k);
      (1/2 x - 1/2)^n Hypergeom([-n, -n], [1], (x+1)/(x-1))
> Sumtohyper(1/2^n*(-1)^k*binomial(n,k)*
binomial(2*n-2*k,n)*x^(n-2*k),k);
      2^(-n) binomial(2n, n) x^n Hypergeom([-1/2 n + 1/2, -1/2 n], [-n + 1/2], 1/x^2).

```

The above computations show that all the given representations of the Legendre polynomials agree.

To give a more advanced example of an application of `Sumtohyper`, we compute the hypergeometric representation of the difference $P_{n+1}(x) - P_n(x)$ of successive Legendre polynomials:

```

> legendreterm:=binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k;
      legendreterm := binomial(n, k) binomial(-n-1, k) (1/2 - 1/2 x)^k
> Sumtohyper(subs(n=n+1, legendreterm)-legendreterm,k);
      (x + x n - 1 - n) Hypergeom([-n, n+2], [2], 1/2 - 1/2 x)

```

We give more examples of how Zeilberger's algorithm can be applied in rather different situations.

The following recurrence equation of the *Apéry numbers*

$$A_n := \sum_{k=0}^n \binom{n}{k}^2 \binom{n+k}{k}^2$$

was an essential tool in Apéry's proof of the irrationality of

$$\zeta(3) = \sum_{j=1}^{\infty} \frac{1}{j^3} :$$

```

> sumrecursion(binomial(n,k)^2*binomial(n+k,k)^2,k,A(n));
      (n+2)^3 A(n+2) - (3+2n)(17n^2+51n+39)A(n+1) + (n+1)^3 A(n) = 0.

```


Dougall's identity

$$(2.2) \quad {}_7F_6 \left(\begin{matrix} a, 1 + \frac{a}{2}, b, c, d, 1 + 2a - b - c - d + n, -n \\ \frac{a}{2}, 1 + a - b, 1 + a - c, 1 + a - d, b + c + d - a - n, 1 + a + n \end{matrix} \middle| 1 \right) = \frac{(1+a)_n (1+a-b-c)_n (1+a-b-d)_n (1+a-c-d)_n}{(1+a-b)_n (1+a-c)_n (1+a-d)_n (1+a-b-c-d)_n}$$

is proven by

```
> sumrecursion(hyperterm([a, 1+a/2, b, c, d, 1+2*a-b-c-d+n, -n],
> [a/2, 1+a-b, 1+a-c, 1+a-d, b+c+d-a-n, 1+a+n], 1, k), k, S(n));
-(a-d+n+1)(n+1+a-c)(n+1-b+a)(-b-c-d+a+n+1)
S(n+1) + (1+a+n)(n+1-c+a-d)(n+1-b+a-d)
(n-c+a+1-b)S(n) = 0.
```

From this result, the right-hand side (2.2) of Dougall's identity can be read off directly. The complete computation is performed by

```
> closedform(hyperterm([a, 1+a/2, b, c, d, 1+2*a-b-c-d+n, -n],
> [a/2, 1+a-b, 1+a-c, 1+a-d, b+c+d-a-n, 1+a+n], 1, k), k, n);
pochhammer(a+1, n) pochhammer(-d-c+1+a, n)
pochhammer(-d+1-b+a, n) pochhammer(a+1-b-c, n) / (
pochhammer(1+a-d, n) pochhammer(1+a-c, n)
pochhammer(1+a-b, n) pochhammer(a-d+1-b-c, n)).
```

Notice how important rational factorization is for such examples.

The *Wilson polynomials* have the representation⁵

$$W_n(x) = {}_4F_3 \left(\begin{matrix} -n, a+b+c+d+n-1, a-x, a+x \\ a+b, a+c, a+d \end{matrix} \middle| 1 \right).$$

They include all classical systems like the Jacobi and Hahn polynomials. We get

```
> sumrecursion(hyperterm([-n, a+b+c+d+n-1, a-x, a+x],
> [a+b, a+c, a+d], 1, k), k, W(n));
(d+a+n+1)(n+1+a+c)(n+b+1+a)(a+2n+c+b+d)
(a+b+c+d+n)W(n+2) - (2n+1+a+b+c+d)(8cdn
+8bdn+bd a^2+8bcn+3b^2n+bc a^2+2b^2d+6anbc
+7dn^2+6bn cd+3d^2n+2cd^2+2bd^2+7cn^2+3c^2n
+2c^2d+2bc^2+7bn^2+2b^2c+7an^2+8adn+4n^3a
+2d^2n^2+4dn^3+2c^2n^2+4cn^3+2b^2n^2+4bn^3+6anbd
+b^2+c^2+8acn+d^2+8abn+4abc+2ab+6ancd
+bcd^2+bc^2d+ab^2d+2ab^2n+acd^2+2ac^2n+abd^2
+ab^2c+abc^2+b^2cd+4abcd+ac^2d+3a^2n+6bn^2a
+6bn^2d+6bn^2c+2b^2nd+2b^2nc+4n^3+2n^4+2c^2nb
+2ad^2+2ac^2+2ab^2+6cn^2a+6cn^2d+2c^2nd+2a^2cn
+2a^2bn+2x^2bd+2x^2cd+2x^2ad+x^2a^2+2x^2bc
```

⁵Sometimes a different standardization is used. But this is not essential.

$$\begin{aligned}
 &+ 2x^2 ac + x^2 d^2 + 2x^2 ab + x^2 b^2 + x^2 c^2 + 4bcd + 2d^2 nc \\
 &+ 2d^2 nb + 2a^2 d + 6dn^2 a + 4abd + 2n^2 + 2a^2 b + 2a^2 c \\
 &+ 2bc + 4acd + 2ad^2 n + a^2 + 2ac + 3bn + 2bd + 2cd \\
 &+ 3dn + 2ad + 2a^2 dn + 3an + 4ax^2 n + 4bx^2 n + 4cx^2 n \\
 &+ 4dx^2 n + 4x^2 n^2 + 2ax^2 + 2bx^2 + 2cx^2 + 2dx^2 + 4x^2 n \\
 &+ 3cn + 2a^2 n^2 + cda^2)W(n+1) + (n+1)(n+d+c)(n+b+d) \\
 &(n+b+c)(d+2n+a+b+c+2)W(n) = 0,
 \end{aligned}$$

a recurrence equation for $W_n(x)$ which, however, is rather complicated since the middle coefficient admits no rational factorization.

One knows from the theory that the recurrence equation has a special form which can be found by the command Sumrecursion:

```
> Sumrecursion(hyperterm([-n,n+a+b+c+d-1,a+x,a-x],
> [a+b,a+c,a+d],1,k),k,W(n,x));
```

$$\begin{aligned}
 (x-a)(a+x)W(n,x) &= ((a+d+n)(n+a+c)(n+b+a) \\
 &(a+b+c+d+n-1)W(n+1,x)) / (\\
 &(a+2n+c+b+d)(2n+b+a+c+d-1)) - (\\
 &\frac{(a+d+n)(n+a+c)(n+b+a)(a+b+c+d+n-1)}{(a+2n+c+b+d)(2n+b+a+c+d-1)} \\
 &+ \frac{(n+d+c-1)(n+b+d-1)(n+b-1+c)n}{(2n+b+a+c+d-1)(2n-2+b+d+a+c)}) \\
 &W(n,x) + \\
 &\frac{(n+d+c-1)(n+b+d-1)(n+b-1+c)nW(n-1,x)}{(2n+b+a+c+d-1)(2n-2+b+d+a+c)}.
 \end{aligned}$$

A similar recurrence equation exists in x :

```
> Sumrecursion(hyperterm([-n,n+a+b+c+d-1,a+x,a-x],
> [a+b,a+c,a+d],1,k),k,W(x,n));
```

$$\begin{aligned}
 (a+b+c+d+n-1)nW(x,n) &= \\
 \frac{1}{2} \frac{(x+d)(x+c)(x+b)(a+x)W(x+1,n)}{(2x+1)x} &- \\
 \left(\frac{1}{2} \frac{(x+d)(x+c)(x+b)(a+x)}{(2x+1)x} + \frac{1}{2} \frac{(x-d)(x-c)(x-b)(x-a)}{(-1+2x)x} \right) & \\
 W(x,n) + \frac{1}{2} \frac{(x-d)(x-c)(x-b)(x-a)W(x-1,n)}{(-1+2x)x}. &
 \end{aligned}$$

Clausen's formula

$${}_2F_1 \left(\begin{matrix} a, b \\ a+b+1/2 \end{matrix} \middle| x \right)^2 = {}_3F_2 \left(\begin{matrix} 2a, 2b, a+b \\ a+b+1/2, 2a+2b \end{matrix} \middle| x \right)$$

gives the cases when the square of a ${}_2F_1$ function is a ${}_3F_2$. The right-hand side is deduced from the left-hand side by

```
> sumrecursion(hyperterm([a,b],[a+b+1/2],x,j)*
> hyperterm([a,b],[a+b+1/2],x,k-j),j,C(k));
```

$$\begin{aligned}
 &-(k+1)(2a+1+2b+2k)(2a+2b+k)C(k+1) \\
 &+ 2x(k+2b)(k+2a)(a+b+k)C(k) = 0
 \end{aligned}$$

computing the coefficient of the Cauchy product. The resulting hypergeometric term summand of the right-hand side can be obtained in one step by⁶

```
> Closedform(hyperterm([a,b],[a+b+1/2],x,j) *
> hyperterm([a,b],[a+b+1/2],x,k-j),j,k);
Hyperterm([2b, 2a, b+a], [a+b+1/2, 2a+2b], x, k)
```

The computation of a specific *Feynman diagram* [7] yields the representation

$$V(\alpha, \beta, \gamma) = (-1)^{\alpha+\beta+\gamma} \cdot \frac{\Gamma(\alpha+\beta+\gamma-d/2)\Gamma(d/2-\gamma)\Gamma(\alpha+\gamma-d/2)\Gamma(\beta+\gamma-d/2)}{\Gamma(\alpha)\Gamma(\beta)\Gamma(d/2)\Gamma(\alpha+\beta+2\gamma-d)(m^2)^{\alpha+\beta+\gamma-d}} \cdot {}_2F_1\left(\begin{matrix} \alpha+\beta+\gamma-d, & \alpha+\gamma-d/2 \\ \alpha+\beta+2\gamma-d \end{matrix} \middle| z\right).$$

Since one is interested to compute this function for $\alpha, \beta, \gamma \in \mathbb{N}$, and since the computation is easy for $\alpha, \beta, \gamma \in \{0, 1\}$, recurrence equations in these variables can be used. Here is one in β :

```
> sumrecursion((-1)^(alpha+beta+gamma) *
> GAMMA(alpha+beta+gamma-d/2) * GAMMA(d/2-gamma) *
> GAMMA(alpha+gamma-d/2) * GAMMA(beta+gamma-d/2) /
> (GAMMA(alpha) * GAMMA(beta) * GAMMA(d/2) *
> GAMMA(alpha+beta+2*gamma-d) * (m^2)^(alpha+beta+gamma-d)) *
> hyperterm([alpha+beta+gamma-d,
> alpha+gamma-d/2], [alpha+beta+2*gamma-d], z, k), k, V(beta));
8 beta m^4 (beta + 1) (alpha + beta + 1 + gamma - d) z V(beta + 2) + 2 beta m^2
(4 gamma + 2 z beta + 2 z - z d + 2 alpha + 2 beta - 2 d) (2 alpha + 2 beta + 2 + 2 gamma - d)
V(beta + 1) +
(2 alpha + 2 beta + 2 gamma - d) (2 gamma - d + 2 beta) (2 alpha + 2 beta + 2 + 2 gamma - d) V(beta)
= 0.
```

Similarly, one obtains recurrence equations in α and γ .

In some instances, Zeilberger's algorithm does not find the recurrence equation of lowest order. Assume, for example, that we want to deduce the right-hand side from the left-hand side of the identity

$$S_n = \sum_{k=0}^n (-1)^k \binom{n}{k} \binom{3k}{n} = (-3)^n.$$

Therefore, by Zeilberger's algorithm we compute a recurrence equation

```
> RE:=sumrecursion((-1)^k*binomial(n,k)*binomial(3*k,n),k,S(n));
RE := 2(3+2n)S(n+2) + 3(7+5n)S(n+1) + 9(n+1)S(n) = 0
```

and apply Petkovšek's algorithm to find its hypergeometric term solutions

```
> rechyper(RE,S(n));
{-3}
```

which gives the term ratio S_{n+1}/S_n of the resulting hypergeometric term.

Here is another application of Petkovšek's algorithm:

```
> RE:=(n+4)*s(n+2)+s(n+1)-(n+1)*s(n)=0;
RE := (n+4)s(n+2) + s(n+1) - (n+1)s(n) = 0
```

⁶The `Closedform` procedure differs from the `closedform` procedure in that the hypergeometric term is not evaluated.

> rehyper(RE, s(n));

$$\left\{ \frac{n+1}{n+3}, -\frac{(5+2n)(n+1)}{(3+2n)(n+3)} \right\}.$$

If continuous variables are involved, one can also compute *holonomic differential equations* for sums by a Zeilberger type algorithm which is implemented in the `sumdiffseq` procedure.

We use some of the series representations of the Legendre polynomials to deduce the corresponding differential equation:

> sumdiffseq(binomial(n,k)*binomial(-n-1,k)*((1-x)/2)^k,k,P(x));

$$-(-1+x)(x+1)\left(\frac{\partial^2}{\partial x^2}P(x)\right) - 2x\left(\frac{\partial}{\partial x}P(x)\right) + P(x)n(n+1) = 0$$

> sumdiffseq(1/2^n*binomial(n,k)^2*(x-1)^(n-k)*(x+1)^k,k,P(x));

$$-(-1+x)(x+1)\left(\frac{\partial^2}{\partial x^2}P(x)\right) - 2x\left(\frac{\partial}{\partial x}P(x)\right) + P(x)n(n+1) = 0.$$

To show a quadratic transformation like

$${}_2F_1\left(\begin{matrix} a, b \\ 2b \end{matrix} \middle| \frac{4x}{(1+x)^2}\right) = (1+x)^{2a} \cdot {}_2F_1\left(\begin{matrix} a, a-b+1/2 \\ b+1/2 \end{matrix} \middle| x^2\right),$$

we prove that both sides satisfy the same differential equation, and show that enough initial values agree:

> sumdiffseq(hyperterm([a,b],[2*b],4*x/(1+x)^2,k),k,Q(x));

$$\begin{aligned} & -x(x-1)(1+x)^2\left(\frac{\partial^2}{\partial x^2}Q(x)\right) + 2(1+x)(-x^2+bx^2-2xa+b)\left(\frac{\partial}{\partial x}Q(x)\right) \\ & + 4Q(x)(x-1)ab = 0 \end{aligned}$$

> sumdiffseq((1+x)^(2*a)*hyperterm([a,a-b+1/2],[b+1/2],x^2,k),k,Q(x));

$$\begin{aligned} & -x(x-1)(1+x)^2\left(\frac{\partial^2}{\partial x^2}Q(x)\right) + 2(1+x)(-x^2+bx^2-2xa+b)\left(\frac{\partial}{\partial x}Q(x)\right) \\ & + 4Q(x)(x-1)ab = 0 \end{aligned}$$

> eval([hypergeom([a,b],[2*b],4*x/(1+x)^2)=

(1+x)^(2*a)*hypergeom([a,a-b+1/2],[b+1/2],x^2),

> diff(hypergeom([a,b],[2*b],4*x/(1+x)^2)=

(1+x)^(2*a)*hypergeom([a,a-b+1/2],[b+1/2],x^2),x],x=0);

$$[1 = 1, 2a = 2a].$$

On p. 258 in Ramanujan's second notebook one finds the identity

$${}_2F_1\left(\begin{matrix} \frac{1}{3}, \frac{2}{3} \\ 1 \end{matrix} \middle| 1 - \left(\frac{1-x}{1+2x}\right)^3\right) = (1+2x) {}_2F_1\left(\begin{matrix} \frac{1}{3}, \frac{2}{3} \\ 1 \end{matrix} \middle| x^3\right).$$

With Garvan [8] we can ask the question: For which A, B, C, a, b, c, d is

$${}_2F_1\left(\begin{matrix} A, B \\ C \end{matrix} \middle| 1 - \left(\frac{1-x}{1+2x}\right)^3\right) = (1+2x)^d {}_2F_1\left(\begin{matrix} a, b \\ c \end{matrix} \middle| x^3\right)?$$

A computation with Maple gives

> first:=hyperterm([A,B],[C],1-((1-x)/(1+2*x))^3,k);

$$first := \frac{\text{pochhammer}(A, k) \text{pochhammer}(B, k) \left(1 - \frac{(1-x)^3}{(1+2x)^3}\right)^k}{\text{pochhammer}(C, k) k!}$$

```

> second:=(2*x+1)^d*hyperterm([a,b],[c],x^3,k);
      second := 
$$\frac{(1+2x)^d \text{pochhammer}(a, k) \text{pochhammer}(b, k) (x^3)^k}{\text{pochhammer}(c, k) k!}$$

> DE1:=sumdiffeq(first,k,S(x));

      DE1 := 
$$x(x-1)(1+x+x^2)(1+2x)^2 \left(\frac{\partial^2}{\partial x^2} S(x)\right) + (1+2x)(4x^4 + 9Bx^3 + 9Ax^3 - 8Cx^3 + 3x^3 - 12Cx^2 + 9Bx^2 + 9Ax^2 + 3x^2 - x - 6Cx + 9Ax + 9Bx - C) \left(\frac{\partial}{\partial x} S(x)\right) + 9(x-1)^2 BAS(x) = 0$$

> DE2:=sumdiffeq(second,k,S(x));

      DE2 := 
$$x(x-1)(1+x+x^2)(1+2x)^2 \left(\frac{\partial^2}{\partial x^2} S(x)\right) + (1+2x)(2x^4 + 6bx^4 - 4dx^4 + 6ax^4 + 3bx^3 + 3ax^3 + x^3 - 6cx + 4x + 4dx + 2 - 3c) \left(\frac{\partial}{\partial x} S(x)\right) + (-12x^4 ad + 36bax^4 + 4d^2 x^4 - 12x^4 bd - 2x^3 d - 6x^3 ad + 36bax^3 - 6x^3 bd + 9bax^2 - 12dx - 4d^2 x + 12cxd + 6cd - 4d)S(x) = 0$$

> DE:=collect(lhs(DE1)-lhs(DE2),[S(x),diff(S(x),x)]);

      DE := 
$$\begin{aligned} & ((1+2x)(4x^4 + 9Bx^3 + 9Ax^3 - 8Cx^3 + 3x^3 - 12Cx^2 + 9Bx^2 + 9Ax^2 + 3x^2 - x - 6Cx + 9Ax + 9Bx - C) - (1+2x)(2x^4 + 6bx^4 - 4dx^4 + 6ax^4 + 3bx^3 + 3ax^3 + x^3 - 6cx + 4x + 4dx + 2 - 3c)) \left(\frac{\partial}{\partial x} S(x)\right) \\ & + (9(x-1)^2 BA + 12x^4 ad - 36bax^4 - 4d^2 x^4 + 12x^4 bd + 2x^3 d + 6x^3 ad - 36bax^3 + 6x^3 bd - 9bax^2 + 12dx + 4d^2 x - 12cxd - 6cd + 4d)S(x) \end{aligned}$$

> firstcoeff:=collect(frontend(coeff,[DE,S(x)]),x);

      firstcoeff := 
$$(12ad - 36ba - 4d^2 + 12bd)x^4 + (2d + 6ad - 36ba + 6bd)x^3 + (-9ba + 9BA)x^2 + (-18BA + 12d + 4d^2 - 12cd)x + 9BA - 6cd + 4d$$

> secondcoeff:=collect(frontend(coeff,[DE,diff(S(x),x)]),x);

      secondcoeff := 
$$(4 - 12b + 8d - 12a)x^5 + (6 + 18B + 18A - 16C + 4d - 12a - 12b)x^4 + (-3b - 3a + 8 + 27B + 27A - 32C)x^3 + (12c - 7 - 8d - 24C + 27B + 27A)x^2 + (-9 - 8C + 9A + 9B - 4d + 12c)x - C - 2 + 3c$$

> LIST:={coeffs(firstcoeff,x)} union {coeffs(secondcoeff,x)};

      LIST := {12ad - 36ba - 4d^2 + 12bd, 9BA - 6cd + 4d, -9ba + 9BA, -18BA + 12d + 4d^2 - 12cd, 6 + 18B + 18A - 16C + 4d - 12a - 12b, 4 - 12b + 8d - 12a, -3b - 3a + 8 + 27B + 27A - 32C, -9 - 8C + 9A + 9B - 4d + 12c, 12c - 7 - 8d - 24C + 27B + 27A, -C - 2 + 3c, 2d + 6ad - 36ba + 6bd}

```

> solve(LIST, {A, B, C, a, b, c, d});

$$\begin{aligned} & \{d = d, B = \frac{1}{3}d, A = \frac{1}{3}d + \frac{1}{3}, b = \frac{1}{3}d, C = \frac{1}{2} + \frac{1}{2}d, a = \frac{1}{3}d + \frac{1}{3}, c = \frac{5}{6} + \frac{1}{6}d\}, \\ & \{d = d, B = \frac{1}{3}d, A = \frac{1}{3}d + \frac{1}{3}, b = \frac{1}{3}d + \frac{1}{3}, C = \frac{1}{2} + \frac{1}{2}d, a = \frac{1}{3}d, \\ & c = \frac{5}{6} + \frac{1}{6}d\}, \{d = d, B = \frac{1}{3}d + \frac{1}{3}, b = \frac{1}{3}d, C = \frac{1}{2} + \frac{1}{2}d, a = \frac{1}{3}d + \frac{1}{3}, \\ & c = \frac{5}{6} + \frac{1}{6}d, A = \frac{1}{3}d\}, \{d = d, B = \frac{1}{3}d + \frac{1}{3}, b = \frac{1}{3}d + \frac{1}{3}, C = \frac{1}{2} + \frac{1}{2}d, \\ & a = \frac{1}{3}d, c = \frac{5}{6} + \frac{1}{6}d, A = \frac{1}{3}d\}. \end{aligned}$$

This leads to the unique solution

$${}_2F_1\left(\begin{matrix} d/3, (1+d)/3 \\ (1+d)/2 \end{matrix} \middle| 1 - \left(\frac{1-x}{1+2x}\right)^3\right) = (1+2x)^d {}_2F_1\left(\begin{matrix} d/3, (1+d)/3 \\ (5+d)/6 \end{matrix} \middle| x^3\right)$$

since the hypergeometric functions are symmetric in their upper parameters. Note that this is exactly the computation from § 1.3. In this problem, a nonlinear system had to be solved.

There is a theory of *basic* hypergeometric (q -hypergeometric) terms A_k for which A_{k+1}/A_k is rational in q^k . For almost all the results and algorithms corresponding q -versions exist.

The corresponding series is called the *basic hypergeometric series*

$${}_r\phi_s\left(\begin{matrix} a_1 & a_2 & \cdots & a_r \\ b_1 & b_2 & \cdots & b_s \end{matrix} \middle| q, x\right) := \sum_{k=0}^{\infty} A_k x^k =$$

$$\sum_{k=0}^{\infty} \frac{(a_1; q)_k \cdot (a_2; q)_k \cdots (a_r; q)_k x^k}{(b_1; q)_k (b_2; q)_k \cdots (b_s; q)_k (q; q)_k} \left((-1)^k q^{\binom{k}{2}}\right)^{1+s-r}$$

where $(a; q)_k := \prod_{j=0}^{k-1} (1 - a q^j)$ is the q -Pochhammer-Symbol. A_k is a q -hypergeometric term and fulfills the recurrence equation ($k \in \mathbb{N}$)

$$A_{k+1} := \frac{(1 - a_1 q^k) \cdots (1 - a_r q^k)}{(1 - b_1 q^k) \cdots (1 - b_s q^k) (1 - q^{k+1})} \cdot A_k$$

with the initial value $A_0 := 1$.

All classical orthogonal polynomial families have at least one, but most families possess many q -analogues.

The q -analogue of Zeilberger's algorithm gives for example for the q -Laguerre polynomials

$$L_n^{(\alpha)}(x; q) = \frac{(q^{\alpha+1}; q)_n}{(q; q)_n} {}_1\phi_1\left(\begin{matrix} q^{-n} \\ q^{\alpha+1} \end{matrix} \middle| q, -xq^{n+\alpha+1}\right),$$

the recurrence equation

> read 'qsum.mpl';

Copyright 1998, Harald Boeing & Wolfram Koepf

Konrad – Zuse – Zentrum Berlin

```
> qsumrecursion(qpochhammer(q^(alpha+1), q, n) / qpochhammer(q, q, n) *
> qphihyperterm([q^(-n)], [q^(alpha+1)], q, -
x*q^(n+alpha+1), k),
> q, k, L(n));
```

$$-q(-1 + q^n)L(n) + (-q^2 + q^{(2n+\alpha)}x - q + q^{(n+\alpha+1)} + q^{(1+n)})L(n-1) + q(q - q^{(\alpha+n)})L(n-2) = 0.$$

The q -analogue of Zeilberger's algorithm generates a third order recurrence equation for the left-hand side of Jackson's q -analogue of Dixon's identity

$$\sum_{k=-n}^n (-1)^k \begin{bmatrix} n+b \\ n+k \end{bmatrix}_q \begin{bmatrix} n+c \\ c+k \end{bmatrix}_q \begin{bmatrix} b+c \\ b+k \end{bmatrix}_q q^{\frac{k(3k-1)}{2}} = \frac{(q; q)_{n+b+c}}{(q; q)_n (q; q)_b (q; q)_c} :$$

```
> term:=(-1)^k*qbinomial(n+b, n+k, q)*qbinomial(n+c, c+k, q)*
> qbinomial(b+c, b+k, q)*q^(k*(3*k-1)/2);
```

$$term := (-1)^k qbinomial(n+b, n+k, q) qbinomial(n+c, c+k, q) qbinomial(b+c, b+k, q) q^{(1/2)k(3k-1)}$$

```
> RE:=qsumrecursion(term, q, k, S(n));
```

$$\begin{aligned} RE := & -(-q^{(2n)} + q)(q^n + 1)(-1 + q^n)q^3 S(n) - (q^5 - q^{(4+n+c+b)} + q^4 \\ & + q^{(3+2n+b)} + q^{(3+2n+c)} - q^{(3+2n)} - q^{(3+c+b+n)} \\ & - q^{(3+n+b)} - q^{(3+n+c)} + q^3 + q^{(2n+b+2)} + q^{(2n+c+2)} \\ & - q^{(2n+c+2+b)} - q^{(3n+c+2)} - q^{(3n+2)} - q^{(3n+b+2)} \\ & + q^{(b+c+2+4n)} - q^{(3n+1)} + q^{(1+4n+c+b)} + q^{(4n+c+b)})q \\ & S(n-1) + (q^{(2+2n)} - q^{(3n+c+b)} + q^{(2n+b+2)} + q^4 \\ & - q^{(4+n+c+b)} - q^{(3n+1+c+b)} - q^{(3+n+b)} + q^{(3+2n+b)} + q^6 \\ & + q^5 - q^{(3n+c+2+b)} - q^{(4+n+b)} + q^{(2n+c+2)} - q^{(3+n+c)} \\ & + q^{(3+2n+c)} - q^{(4+n+c)})(q - q^{(n+c+b)})S(n-2) - \\ & (q - q^{(n+c+b)})(q^2 - q^{(n+b)})(q^2 - q^{(n+c)})(q^2 - q^{(n+c+b)}) \\ & S(n-3) = 0. \end{aligned}$$

The q -analogue of Petkovšek's algorithm decides whether a q -holonomic recurrence equation has q -hypergeometric term solutions. It finds the right-hand side of the q -analogue of Dixon's identity:

```
> qrecsolve(RE, q, S(n));
[[ (q^(n+1) - 1)S(n+1) + (-q^(1+c+b+n) + 1)S(n) = 0 ]].
```

In many cases much simpler is Paule's *creative symmetrizing*. With this method, we symmetrize the summand, and get the result in one step.

```
> M:=qsimpcomb(subs(k=-k, term)/term);
```

$$M := q^k$$

```
> qsumrecursion((1+M)/2*term, q, k, S(n));
```

$$(1 - q^n)S(n) + (q^{(n+c+b)} - 1)S(n-1) = 0.$$

There exist similar algorithms for definite integration instead of summation.

The *Bateman integral representation*

$$\int_0^1 t^{c-1} (1-t)^{d-1} {}_2F_1 \left(\begin{matrix} a, b \\ c \end{matrix} \middle| tx \right) dt = \frac{\Gamma(c)\Gamma(d)}{\Gamma(c+d)} {}_2F_1 \left(\begin{matrix} a, b \\ c+d \end{matrix} \middle| x \right)$$

is proven by

```

> intrecursion(t^(c-1)*(1-t)^(d-1)*
> hyperterm([a,b],[c],t*x,k),t,B(k));
      -(k+1)(k+d+c)B(k+1)+B(k)x(b+k)(a+k)=0
> assume(d>0,c>0);
> init:=int(t^(c-1)*(1-t)^(d-1),t=0..1);
      init:=B(c~,d~).

```

Here $B(c, d)$ denotes the *Beta function*

$$B(c, d) = \frac{\Gamma(c)\Gamma(d)}{\Gamma(c+d)}.$$

As part of the q -Askey-Wilson scheme [15] we have the *Askey-Wilson polynomials*

$$p_n(x; a, b, c, d|q) = (ab; q)_n (ac; q)_n (ad; q)_n a^{-n}.$$

$${}_4\phi_3 \left(\begin{matrix} q^{-n}, abcdq^{n-1}, ae^{i\theta}, ae^{-i\theta} \\ ab, ac, ad \end{matrix} \middle| q, q \right).$$

The connection between those families can be written as

$$p_n(x; \alpha, \beta, \gamma, d|q) = \sum_{m=0}^n C_m(n) p_m(x; a, b, c, d|q),$$

with *connection coefficients* $C_m(n)$.

Askey and Wilson developed the following representation for $C_m(n)$:

$$C_m(n) = \frac{(\alpha d, \beta d, \gamma d, q; q)_n (\alpha \beta \gamma d q^{n-1}; q)_m}{(\alpha d, \beta d, \gamma d, q, abcdq^{m-1}; q)_m (q; q)_{n-m}} q^{m^2-nm} d^{m-n} \cdot {}_5\phi_4 \left(\begin{matrix} q^{m-n}, \alpha \beta \gamma d q^{n+m-1}, adq^m, bdq^m, cdq^m \\ abcdq^{2m}, \alpha d q^m, \beta d q^m, \gamma d q^m \end{matrix} \middle| q, q \right).$$

In the general case this is *not* a q -hypergeometric term.

However, for the special case $\beta = b$ and $\gamma = c$ we get by the q -Zeilberger algorithm ([3], compare [9], Sections 7.5 and 7.6)

$$C_m(n) = \frac{(\alpha/a; q)_{n-m} (abcdq^{n-1}; q)_m (q, bc, bd, cd; q)_n a^{n-m}}{(q, bc, bd, cd; q)_m (abcdq^{m-1}; q)_m (q, abcdq^{2m}; q)_{n-m}} :$$

```

> c:='c': d:='d':
> term:=qpochhammer(alpha*d,beta*d,gamma*d,q,q,n)*
> qpochhammer(alpha*beta*gamma*d*q^(n-1),q,m)/
> qpochhammer(alpha*d,beta*d,gamma*d,q,a*b*c*d*q^(m-1),q,m)/
> qpochhammer(q,q,n-m)*q^(m^2-n*m)*d^(m-n)*

```



```

> qphihyperterm([q^(m-n), alpha*beta*gamma*d*q^(n+m-1),
> a*d*q^m, b*d*q^m, c*d*q^m],
> [a*b*c*d*q^(2*m), alpha*d*q^m, beta*d*q^m, gamma*d*q^m], q, q, j) :
> qsumrecursion(subs({beta=b, gamma=c}, term), q, j, C(m)) ;

```

$$\begin{aligned}
 & -(-1 + q^m)(-q + cdq^m)(caq^{(n+m)}bd - q)(bcq^m - q)(-q + bdaq^m) \\
 & (-q^3 + abcdaq^{(2m)})(aq^m - \alpha q^n)C(m) + (-q^2 + caq^m bd) \\
 & (-q + abcdaq^{(2m)})q^2(-q^m + q^{(1+n)})(-q^2 + \alpha cq^{(n+m)}bd) \\
 & C(m-1) = 0
 \end{aligned}$$

and similar results for $\alpha = a, \gamma = c$ and for $\alpha = a, \beta = b$:

From these results one can derive the connection coefficients between many families of the q -Askey-Wilson tableau by limit computations.

REFERENCES

- [1] E. BEKE, *Die Irreducibilität der homogenen linearen Differentialgleichungen*, Math. Ann. 45 (1894), pp. 278–294.
- [2] ———, *Die symmetrischen Functionen bei linearen homogenen Differentialgleichungen*, Math. Ann. 45 (1894), pp. 295–300.
- [3] H. BÖING AND W. KOEPF, *Algorithms for q -hypergeometric summation in computer algebra*, J. Symbolic Comput. (1999), to appear.
- [4] I. N. BRONSHTEIN AND K. A. SEMEDYAYEV, *Handbook of Mathematics*, 3rd edition, Springer, New York, 1997.
- [5] M. BRONSTEIN, *Symbolic Integration I: Transcendental Functions*, Springer, Berlin, 1997.
- [6] B. W. CHAR ET AL., *Maple V Language Reference Manual*, Springer, New York, 1991.
- [7] J. FLEISCHER AND O. V. TARASOV, *Calculation of Feynman diagrams from their small momentum expansion*, Z. Phys. C64 (1994), pp. 413–425.
- [8] F. G. GARVAN, *Ramanujan's theories of elliptic functions to alternative bases—a symbolic excursion*, J. Symbolic Comput. 20 (1995), pp. 517–536.
- [9] G. GASPER AND M. RAHMAN, *Basic Hypergeometric Series*, Encyclopedia of Mathematics and its Applications, 35, Cambridge University Press, London and New York, 1990.
- [10] K. O. GEDDES, S. R. CZAPOR AND G. LABAHN, *Algorithms for computer algebra*, Kluwer Academic Publ., Boston/Dordrecht/London, 1992.
- [11] R. W. GOSPER JR., *Decision procedure for indefinite hypergeometric summation*, Proc. Natl. Acad. Sci. USA 75 (1978), pp. 40–42.
- [12] D. GRUNTZ AND W. KOEPF, *Maple package on formal power series*, Maple Technical Newsletter 2 (2) (1995), pp. 22–28.
- [13] A. C. HEARN, *Reduce User's Manual, Version 3.6*, RAND Co., Santa Monica, CA, 1995.
- [14] R. D. JENKS AND R. S. SUTOR, *AXIOM. The Scientific Computation System*, Springer, Berlin, 1993.
- [15] R. KOEKOEK AND R. F. SWARTTOUW, *The Askey-scheme of hypergeometric orthogonal polynomials and its q analogue*, Report 94–05, Technische Universiteit Delft, Faculty of Technical Mathematics and Informatics, Delft, 1994; updated electronic version available at <http://aw.twi.tudelft.nl/~koekoek/research.html>.
- [16] W. KOEPF, *W, Power series in Computer Algebra*, J. Symbolic Comput. 13 (1992), pp. 581–603.
- [17] ———, *Hypergeometric Summation*, Vieweg, Braunschweig/Wiesbaden, 1998.
- [18] T. H. KOORNWINDER, *On Zeilberger's algorithm and its q -analogue: a rigorous description*, J. Comput. Appl. Math. 48 (1993), pp. 91–111.
- [19] MACSYMA: REFERENCE MANUAL, Macsyma, Inc., Arlington, MA 02174, USA.
- [20] Y.-K. MAN AND F. J. WRIGHT, *Fast polynomial dispersion computation and its application to indefinite summation*, Proc. of ISSAC 94, ACM Press, New York, (1994), pp. 175–180.
- [21] A. W. OVERHAUSER AND Y. I. KIM, *Problem 94–2*, SIAM Review 36 (1994), p. 107.
- [22] A. RICH, J. RICH AND D. STOUTEMYER, *DERIVE User Manual*, Soft Warehouse, Inc., 3660 Waialae Avenue, Suite 304, Honolulu, Hawaii, 96816-3236.
- [23] B. SALVY AND P. ZIMMERMANN, *GFUN: A package for the manipulation of generating and holonomic functions in one variable*, ACM Trans. Math. Software 20 (1994), pp. 163–177.
- [24] R. P. STANLEY, *Differentiably finite power series*, European J. Combin. 1 (1980), pp. 175–188.
- [25] ST. WOLFRAM, *The Mathematica Book*, Wolfram Media, Champaign, Illinois, and Cambridge University Press, Cambridge, 1996.